Conference in Honour of André Galligo

on the Occasion of his 60th Birthday

Nice, June 2006

Computational Aspects in the Theory of Singularities

Gert-Martin Greuel

University of Kaiserslautern

Mathematisches Forschungsinstitut Oberwolfach

(Germany)

1. Historical Remarks and a Counterexample
2. A Theorem in Group Theory
3. Curves and Surfaces with many Singularities

Historical remarks and a counterexample

The birth of SINGULAR can be dated back to about 1982, when G. Pfister and I tried to generalize the following **theorem of K. Saito**:

Let $(X,0)$ be the germ of an isolated hypersurface singularity. The following conditions are equivalent.

(1) $(X,0)$ is quasi–homogeneous.

(2) $\mu(X,0) := \dim_{\mathbb{C}} \mathbb{C}[[x_1,\ldots,x_n]]/\langle \frac{\partial f}{\partial x_1},\ldots,\frac{\partial f}{\partial x_n}\rangle =$
$\tau(X,0) := \dim_{\mathbb{C}} \mathbb{C}[[x_1,\ldots,x_n]]/\langle f, \frac{\partial f}{\partial x_1},\ldots,\frac{\partial f}{\partial x_n}\rangle$

(3) The Poincaré complex $0 \to \mathbb{C} \to \mathcal{O}_{X,0} \to \Omega^1_{X,0} \to \ldots \to \Omega^n_{X,0} \to 0$, of $(X,0)$ is exact.

It was conjectured that a similar theorem should hold for complete intersections.

If $(X,0)$ is the germ of a curve singularity we succeeded in proving the equivalence of (1) and (2).

To understand the relationship with (3) we first translated the question about exactness of the Poincaré complex into a purely algebraic question (note that the differential is only $\mathbb{C}$–linear but not $\mathcal{O}_{X,0}$–linear). Then we tried to compute examples which turned out to be rather difficult by hand. In those days there was no computer algebra system available which could compute Milnor numbers and Tjurina numbers for non–trivial examples. Such a system would have required the implementation of algorithms for computing **standard bases for ideals and modules over local rings**. Let us consider the following example.

Let $f = xy + z^4, g = xz + y^5 + yz^2$ and $(X,0) \subset (\mathbb{C}^3,0)$ be defined by $f = g = 0$. In this case we have,

$$\begin{aligned}
\mu(X,0) &= \dim_{\mathbb{C}} \mathbb{C}\,[[x,y,z]]/\langle f, M_1, M_2, M_3\rangle - \dim_{\mathbb{C}} \mathbb{C}[[x,y,z]]/\langle \tfrac{\partial f}{\partial x}, \tfrac{\partial f}{\partial y}, \tfrac{\partial f}{\partial z}\rangle,\\
\tau(X,0) &= \dim_{\mathbb{C}} \mathbb{C}[[x,y,z]]/\langle f, g, M_1, M_2, M_3\rangle,
\end{aligned}$$

where M_1, M_2, M_3 are the 2–minors of the Jacobian matrix of (f,g).

In SINGULAR we can compute these numbers as follows:

```
> ring R = 0, (x,y,z), ds;    // localisation Q[x,y,z]_<x,y,z>
> poly f, g = xy+z4, xz+y5+yz2;
> ideal I = f, g;
> matrix J = jacob(I);         // Jacobian matrix
> ideal Tjur = I, minor(J,2);
> vdim(std(Tjur));             // compute K-dimension of R/Tjur
12                             // the Tjurina number is 12
```

Alternatively, we can use the built-in command `tjurina` from `sing.lib`.

```
> LIB "sing.lib";    // load the library sing.lib
> tjurina(I);
12
```

It is known that for quasihomogeneous complete intersections Tjurina and Milnor number coincide.

Computing the Milnor number we see that $(X, 0)$ is not quasihomogeneous:

```
> milnor(I);                          // from sing.lib
13                                    // the Milnor number is 13
```

However, the Poincaré complex is exact. To see this, we showed that it suffices to check that $\mu(X,0) = \dim_{\mathbb{C}} \Omega^2_{X,0} - \dim_{\mathbb{C}} \Omega^3_{X,0}$. Note that $\dim_{\mathbb{C}} \Omega^3_{X,0} = 1$.

$$\Omega^2_{X,0} = \Omega^2_{\mathbb{C}^3,0} / \Big(\langle f, g \rangle \Omega^2_{\mathbb{C}^3,0} + df \wedge \Omega^1_{\mathbb{C}^3,0} + dg \wedge \Omega^1_{\mathbb{C}^3,0} \Big)$$

is isomorphic to $\mathcal{O}^3_{X,0}/M$, where $M \subset \mathcal{O}^3_{X,0}$ is generated by the six vectors

$$\left(\tfrac{\partial f}{\partial y}, \tfrac{\partial f}{\partial z}, 0\right),\ \left(\tfrac{\partial f}{\partial x}, 0, -\tfrac{\partial f}{\partial z}\right),\ \left(0, \tfrac{\partial f}{\partial x}, \tfrac{\partial f}{\partial y}\right),\ \left(\tfrac{\partial g}{\partial y}, \tfrac{\partial g}{\partial z}, 0\right),\ \left(\tfrac{\partial g}{\partial x}, 0, -\tfrac{\partial g}{\partial z}\right),\ \left(0, \tfrac{\partial g}{\partial x}, \tfrac{\partial g}{\partial y}\right):$$

```
> qring Q = std(I);        // quotient ring Q=R/I
> poly f = imap(R,f);      // map f from R to Q
> poly g = imap(R,g);
> module M = [diff(f,y),diff(f,z),0], [diff(f,x),0,-diff(f,z)],
.            [0,diff(f,x),diff(f,y)], [diff(g,y),diff(g,z),0],
.            [diff(g,x),0,-diff(g,z)],[0,diff(g,x),diff(g,y)];
> vdim(std(M));
14
```

Thus we computed $\dim_{\mathbb{C}} \Omega^2_{X,0} = 14 = \mu(X,0) + \dim_{\mathbb{C}} \Omega^3_{X,0}$ showing that the Poincaré complex is exact.

The first version of a standard basis algorithm (called `BuchMora`) was implemented in `BASIC` on a ZX–Spectrum by K.P. Neudendorf (born Schemmel) and G. Pfister in 1983. This implementation allowed us to compute first examples.

A serious development started in 1984 with an implementation of Mora's tangent cone algorithm in `Modula-2` on an Atari computer at the Humboldt-University in Berlin (by G. Pfister and a group of students, including Hans Schönemann). After a while, a list of counter-examples to the above mentioned conjecture was produced.

At that time, the system could only compute with coefficients in a small prime field $\mathbb{F}_p$. However, the experiments showed which examples are candidates for a counter-example and how the computations in characteristic 0 should look like. The proof of the following was then given manually.

There are infinitely many counterexamples to the conjectured generalization of Saito's theorem.

A Theorem in Group Theory

While the previous application of Singular was an early example of a nowadays standard application of computer algebra, the following example is rather amazing. **The problem is formulated in purely group–theoretic terms.**

We first translated it into a problem in algebraic respectively arithmetic geometry, where we had to show the existence of rational points on explicitly given varieties defined over finite fields. To solve the problem we had to apply the well–known Hasse–Weil formula but also sophisticated versions of the Lefschetz trace formula as conjectured by Deligne and proved by Fujiwara.

To apply the Hasse–Weil, respectively the Lefschetz trace formula we had to study the geometric structure of certain algebraic varieties given by explicit equations, find their irreducible components, their singular loci, etc.

All this was done by using SINGULAR as an indispensable tool. The hardest part was finally to show that the varieties we ended up with were irreducible over the algebraic closure of given finite fields. But SINGULAR was not only used for these computations it was also essential in finding the correct formulation of the theorem.

As we shall see, parts of the theorem can now be proved without a computer while other parts (in particular the Suzuki groups) still require SINGULAR computations. However, since we give explicit solutions, the correctness of the statements can be verfied by simple (but lengthy) computations either by hand or (better) by any other computer algebra system.

The diversity of the methods required the collaboration of six authors from different fields. The final proof may be considered as an example of the unity of mathematics in our more and more specializing discipline.

The problem in group theory was to **characterize the finite solvable groups by two–variable identities** (like $xy = yx$ for abelian groups) as we explain now.

If G is a group and x, $y \in G$, we inductively define

$$e_1(x,y) := x^{-2}y^{-1}x,\ \ e_{n+1}(x,y) := [xe_n(x,y)x^{-1}, ye_n(x,y)y^{-1}]$$

where the commutator of g, $h \in G$ is defined by $[g,h] := ghg^{-1}h^{-1}$.

The following theorem was proved by T. Bandman, G.-M. Greuel, F. Grunewald, B. Kunyavski, G. Pfister, and E. Plotkin [Compositio Math. 2006]:

Theorem: A finite Group G is solvable if and only if there is an $n \in \mathbb{N}$ such that $e_n(x,y) = 1$ for all x, $y \in G$.

We start with the classification of the **minimal finite non–solvable groups** G (that is, all subgroups of G are solvable) by J. Thompson in 1968:

1. $\mathrm{PSL}(2,p)$, p a prime number, $p = 5$ or $p > 5$ and $p = \pm 2 \mod 5$.
2. $\mathrm{PSL}(2,2^n)$, $n \geq 2$, a prime number.
3. $\mathrm{PSL}(2,3^n)$, n odd, a prime number.
4. $\mathrm{PSL}(3,3)$.
5. $Sz(2^n)$, n odd.

Since it is easy to see that the finite solvable groups satisfy the proposed identity it is enough to show that for each group G in Thompson's list we have $x, y \in G$

with $e_1(x,y) = e_2(x,y)$ and $y \neq x^{-1}$. By the structure of the sequence e_n, this implies $1 \neq e_1(x,y) = e_n(x,y)$ for all n.

We shall give an **idea on how to prove the theorem** for the group PSL(2,q). The case PSL(3,3) is easy but the case of the Suzuki groups $Sz(2^n)$ is much more difficult.

Proposition: If $q = p^k$ for a prime p and $q \neq 2, 3$, then there are x, y in PSL$(2, \mathbb{F}_q)$ with $y \neq x^{-1}$ and $e_1(x,y) = e_2(x,y)$.

The proof will use some explicit computations with the following matrices. Let $R = \mathbb{Z}$ or $\mathbb{F}_q$ and define

$$x(t) := \begin{pmatrix} t & -1 \\ 1 & 0 \end{pmatrix}, \quad y(b,c) := \begin{pmatrix} 1 & b \\ c & 1+bc \end{pmatrix} \in \mathrm{SL}(2,R)$$

for $t, b, c \in R$.

Let $I \subseteq \mathbb{Z}[b,c,t]$ be the ideal generated by the four entries of the matrix $e_1(x,y) - e_2(x,y)$.

Using SINGULAR we can obtain I as follows:

```
>LIB"linalg.lib";
```

```
>ring R = 0,(b,c,t),dp;

>matrix X[2][2] =     t, -1,
                      1, 0;
>matrix Y[2][2] =     1, b,
                      c, 1+bc;
>matrix iX = inverse(X);
>matrix iY = inverse(Y);
>matrix M=iX*Y*iX*iY*X*X-Y*iX*iX*iY*X*iY;
>ideal I=flatten(M); I;
I[1]=b3c2t2+b2c2t3-b2c2t2-bc2t3-b3ct+b2c2t+b2ct2+2bc2t2+bct3
     +b2c2+b2ct+bc2t-bct2-c2t2-ct3-b2t+bct+c2t+ct2+2bc+c2+bt
     +2ct+c+1;
I[2]=-b3ct2-b2ct3+b2c2t+bc2t2+b3t-b2ct-2bct2-b2c+bct+c2t+ct2
      -bt-ct-b-c-1;
I[3]=b3c3t2+b2c3t3-b2c2t3-bc2t4-b3c2t+b2c3t+2b2c2t2+2bc3t2
     +2bc2t3+b2c2t+2b2ct2+bc2t2-c2t3-ct4-2b2ct+bc2t+c3t+bct2
     +2c2t2+ct3-b2c-b2t+bct+c2t +bt2+3ct2+bc-bt-b-c+1;
```

```
I[4]=-b3c2t2-b2c2t3+b2c2t2+bc2t3+b3ct-b2c2t-b2ct2-2bc2t2-bct3
    -2b2ct+c2t2+ct3+b2t-bct-c2t-ct2+b2-bt-2ct-b-t+1;
```

To prove the Proposition above, it is enough to prove the following

Lemma: Let q be as in the Proposition, then the variety $V(I) \subset \overline{\mathbb{F}}_q^3$ is a curve. The set of $\mathbb{F}_q$–rational points $V^{(q)} = V(I) \cap \mathbb{F}_q^3$ is not empty.

We apply the **theorem of Hasse–Weil** as generalised by Aubry and Perret to singular curves and use the fact that the affine curve C has, at most, $\deg(\overline{C})$ rational points less than the projective closure $\overline{C}$:

Theorem: Let $C \subseteq \mathbb{A}^n$ be an absolutely irreducible affine curve defined over the finite field $\mathbb{F}_q$ and $\overline{C} \subset \mathbb{P}^n$ the projective closure, then the number of $\mathbb{F}_q$–rational points of C is at least $q + 1 - 2p_a\sqrt{q} - d$ with d the degree and p_a the arithmetic genus of $\overline{C}$.

Note that the Hilbert function of $\overline{C}$, $H(t) = dt - p_a + 1$, can be computed from the homogeneous ideal I_h of $\overline{C}$, hence we can compute d and p_a without any knowledge about the singularities of $\overline{C}$.

Let L be the algebraic closure of $\mathbb{F}_q$. To apply the proposition, we have to prove that C is absolutely irreducible, that is, that $IL[b,c,t]$ is a prime ideal. This is already hard to compute. It turned out that the computation over the function field $L(t)$ was easier.

Lemma: $IL(t)[b,c] = \langle f_1, f_2 \rangle$ with

$$
\begin{aligned}
f_1 &= t^2b^4 - t^3(t-2)b^3 + (-t^5+3t^4-2t^3+2t+1)b^2 \\
&\quad + t^2(t^2-2t-1)(t-2)b + (t^2-2t-1)^2 \\
f_2 &= t(t^2-2t-1)c + t^2b^3 + (-t^4+2t^3)b^2 + (-t^5+3t^4-2t^3+2t+1)b \\
&\quad + (t^5-4t^4+3t^3+2t^2)\,.
\end{aligned}
$$

Moreover, we have $IL[b,c,t] = \langle f_1, f_2 \rangle : h^2,\ h = t(t^2-2t-1).$

This can be tested in Singular as follows:

```
>ring S=(0,t),(c,b),lp;
>ideal I=imap(R,I);
>ideal J=std(I); J;
J[1]=(t2)*b4+(-t4+2t3)*b3+(-t5+3t4-2t3+2t+1)*b2+(t5-4t4+3t3+2t2)
     *b+(t4-4t3+2t2+4t+1)
J[2]=(t3-2t2-t)*c+(t2)*b3+(-t4+2t3)*b2+(-t5+3t4-2t3+2t+1)
     *b+(t5-4t4+3t3+2t2)
```

Now $IL(t)[b,c] \cap L[b,c,t] = \langle f_1, f_2 \rangle : h^2 = IL[b,c,t]$. Therefore, it is enough to prove that $IL(t)[b,c]$ is a prime ideal which is equivalent to prove that f_1 is irreducible in $L(t)[b]$. By the lemma of Gauß we have to prove that f_1 is irreducible in $L[t,b]$.

Let $P(x) := t^2 f_1(x/t)$, then

$$\begin{aligned} P(x) &= x^4 - t^2(t-2)x^3 + (-t^5 + 3t^4 - 2t^3 + 2t + 1)x^2 + t^3(t-2)(t^2-2t-1)x \\ &\quad + t^2(t^2-2t-1)^2 . \end{aligned}$$

Clearly it suffices to prove that $P \in L[x,t]$ is irreducible.

To show that P is not divisible by any factor of degree 2 in x we make the following **"Ansatz"**:

$$(*) \qquad p = (x^2 + ax + b)\,(x^2 + gx + d),$$

a, b, g, d polynomials in t with indeterminates `a(i), b(i), g(i), d(i)` as coefficient. It is easy to see that we can assume

$$\deg(b) \leq 5, \deg(a) \leq 3, \deg(d) \leq 3, \deg(g) \leq 2\,.$$

Then a decomposition (*) with `a(i), b(i), g(i), d(i)` $\in \overline{\mathbb{F}}_p$ does not exist if and only if the ideal `C` of the coefficients in x, t of $P-(x^2+ax+b)(x^2+gx+d)$ has no solution in $\overline{\mathbb{F}}_p$. This is equivalent to the fact that a Gröbner basis of `C` contains $1 \in \mathbb{F}_p$.

The ideal `C` of coefficients from our Ansatz:

```
C[1]=-b(5)*d(3)
C[2]=-b(5)*g(2)
C[3]=-b(4)*d(3)-b(5)*d(2)
C[4]=-b(4)*g(2)-b(5)*g(1)-d(3)-1
C[5]=-b(3)*d(3)-b(4)*d(2)-b(5)*d(1)+1
```

```
C[6]=-b(5)-g(2)-1
C[7]=a(0)*b(5)-a(2)*d(3)-b(3)*g(2)-b(4)*g(1)-d(2)+4
```

$\vdots$

```
C[24]=-a(0)^2*b(0)+b(0)^2-b(0)
```

For a given prime p it is easy to compute the Gröbner basis of C and to verify that $1 \in C$. However, we cannot check infinitely many primes. What we do is to use that the polynomials generating C have integer coefficients.

Hence, if we can express some integer m as a polynomial combination of the generators of C where all polynomials have integer coefficients, then for any prime p, $p \nmid m, 1 \in \mathbb{F}_p$ is contained in C (mod p).

We use the **lift** command of SINGULAR to show that (over $\mathbb{Z}$) $m = 4 \in \texttt{C}$:

```
>matrix M=lift(C,4); M;
M[1,1]=-a(0)+8*b(0)*b(3)-8*b(0)*b(4)-16*b(0)*g(1)*g(2)-...
M[2,1]=-a(0)^2+6*a(0)*b(3)-30*a(0)*b(5)*d(1)+200*a(0)*b(5)*d(2)-...
M[3,1]=-8*b(0)*g(1)-8*b(0)*g(2)+8*b(1)*g(2)+8*b(1)-...
M[4,1]=-16*b(0)*g(2)*d(3)-18*b(0)*g(2)+8*b(0)*d(2)-8*b(0)*d(3)-...
M[5,1]=8*a(2)*b(0)+142*a(2)*d(1)*d(3)+41*a(2)*d(1)-...
M[6,1]=a(0)^2*g(2)+8*a(0)*b(0)*d(3)-6*a(0)*b(3)*g(2)+5*a(0)*b(3)+...
M[7,1]=8*b(0)*d(3)+5*b(3)-15*b(5)*d(1)+100*b(5)*d(2)-...

⋮

M[24,1]=0
```

The computation shows that

$\binom{*}{*}$ $$4 = \sum_{i=1}^{24} M[i,1] \cdot \texttt{C}[i].$$

Note that it is difficult to find the polynomials $M[i,j]$ but once they are found it is easy to check that the relation $\binom{*}{*}$ holds.

$\binom{*}{*}$ implies that over $\overline{\mathbb{F}_p}$, $p \neq 2$, P has no quadratic factor. Similarly, it has no linear factor. Thus P is absolutely irreducible in $\mathbb{F}_p[t, x]$ for all $p \neq 2$ ($p = 2$ is treated by a direct computation).

Now we can apply the theorem of Hasse–Weil to prove Lemma 7.3.

We compute the Hilbert polynomial $H(t)$ of the projective curve corresponding to I. We obtain $H(t) = 10t - 11$. The corresponding SINGULAR session is:

```
>ring S=0,(b,c,t,w),dp;
>option(contentSB);
>ideal I=imap(R,I);
>ideal J=std(I); J;
J[1]=bct-t2+2t+1
J[2]=bt3-ct3+t4-b2t+bct-c2t-2bt2+2ct2-3t3+bc+t2+t+1
J[3]=b2c2-b2ct+bc2t-bct2+b2+2bc+c2-b+c-t+2
J[4]=c2t3-ct4+c3t-2c2t2+3ct3-t4-bc2+bt2-2ct2+4t3-2bt+ct-3t2-b-2t
```

It can easily be seen that J induces a Gröbner basis in $\mathbb{F}_p[b, c, t, w]$ for all p, because `option(contentSB)` forces SINGULAR to avoid division by integers.

We homogenise J with respect to w and obtain again a Gröbner basis, with respect to the degree reverse lexicographical ordering. Since the leading coefficients of J have all coefficient 1 and since J and the leading ideal of J have the same Hilbert polynomial, the Hilbert polynomial is the same in any characteristic.

```
J=homog(J,w);
hilbPoly(J);
-11,10                          // H(t) = 10t - 11
```

From the the result we see that the degree $d = 10$ and the arithmetic genus $p_a = 12$. Using theorem 7.4, we obtain:

$$\#V^{(q)} \geq q + 1 - 24\sqrt{q} - 10.$$

This implies that $V^{(q)}$ **is not empty if** $q > 593$.

For the remaining prime powers q, we check directly by computer that $V^{(q)}$ is not empty.

Curves and Surfaces with many Singularities

We describe now a typical example how SINGULAR was used to **support research in algebraic geometry by creating interesting examples**.

Let $X \subset \mathbb{P}^n_{\mathbb{C}}$ be a projective hypersurface being the zero set of $f(z_0, \ldots, z_n) \in \mathbb{C}[z_0, \ldots, z_n]$, a homogeneous polynomial of degree $d > 0$.

Bezout's theorem: If the intersection of n hypersurfaces in $\mathbb{P}^n$ consist of finitely many points then the number of intersection points (counted with appropriate multiplicities) is equal to the product of the degrees of the hypersurfaces.

In particular, if p is a singular point of X and if L is a line in general position then the intersection number of X and L at p is equal to the multiplicity mult (X, p), hence X cannot have any singularity of multiplicity bigger than its degree. To get an estimate for the number of singularities we can use another local invariant, the Milnor number $\mu(X, p) = \dim_{\mathbb{C}} \mathbb{C}\,[[x_1, \ldots, x_n]]\,/\langle \frac{\partial f}{\partial x_1}, \ldots, \frac{\partial f}{\partial x_n}\rangle$

($x_1, \dots, x_n$ local coordiantes and $f = 0$ a local analytic equation of X). If X has only isolated singularities, then $\mu(X,p) < \infty$ for all $p \in X$ and, by choosing general projective coordinates, we may assume that no singularity of X lies on $\{z_0 = 0\}$. Considering the intersection of the hypersurfaces $\frac{\partial f}{\partial z_i} = 0, i = 1, \dots, n$, we obtain from Bezout's theorem **the following necessary condition for the existence of X**.

$$(d-1)^n \geq \sum_{p \in \mathrm{Sing}(X)} \mu(X,p). \qquad (*)$$

Since $\mu(X,p) = 0$ if p is nonsingular and $\mu(X,p) = 1$ iff p is a node we get that the number of singularities of X is bounded by $d^n + O(d^{n-1})$ and that the number of non–nodes is bounded by $\frac{1}{2}d^n + O(d^{n-1})$.

(An A_k–**singularity** has the local analytic equation $x_1^2 + \cdots + x_n^{k+1} = 0$. A_1–singularities are called ***nodes***, A_2-singularities ***cusps***.)



Figure 1: A 4–nodal plane curve of degree 5, with equation $x^5 - \frac{5}{3}x^3 + \frac{5}{16}x - \frac{1}{4}y^3 + \frac{3}{16}y = 0$, which is a deformation of $E_8 : x^5 - y^3 = 0$.



Figure 2: A plane curve of degree 5 with 5 cusps, the maximal possible number. The equation is $\frac{129}{8}x^4y - \frac{85}{8}x^2y^3\frac{57}{32}y^5 - 20x^4 - \frac{21}{4}x^2y^2\frac{33}{8}y^4 - 12x^2y + \frac{73}{8}y^3 + 32x^2 = 0$.

From the very beginning of algebraic geometry, the **existence of hypersurfaces with many singularities has been a problem of constant importance and interest**, from Descartes, Pascal, Newton over Plücker and Severi to Zariski and Harris until nowadays.

Except for the simplest case, the number of nodes on a plane curve settled by Severi in 1921, no general answer is known. The problem turned out to be extremely hard and the partial results so far suggest that a generel condition for the existence of singularities of a given type which is necessary and sufficient at the same time cannot be expected for more complicated singularities than nodes.

Two directions of research have been established in this connection:

(I) to find sufficient existence conditions which are proper (i.e. have the asymptotic $\alpha d^n + O(d^{n-1})$ with a constant α which is not necesarily optimal) or

(II) to find necessary and sufficient conditions for small d and the simplest singularities like nodes and cusps.

Let us first consider (I).

The first general asymptotic proper conditions for the existence were found only in 1989 in the case of plane curves by G.-M. Greuel, Ch. Lossen, E. Shustin (Inventiones Math. 1989):

Theorem: **For any $d \geq 1$ and topological types $S_1, \ldots, S_n$ of plane curve singularities s.t.**

$$\sum_{i=1}^{n} \mu(S_i) \leq \frac{1}{392} d^2$$

there exists an irreducible plane curve of degree d having exactly $S_1, \ldots, S_n$ as singularities.

The coefficient $\alpha = \frac{1}{392}$ has been improved subsequently (cf. our forthcoming book).

This result is just an existence statement, the proof gives no hint how to produce any equation.

Having a method for constructing curves of low degree with many singularities,**Lossen** was able to **produce explicit equations**. In order to check his construction and improve the results, he made extensive use of SINGULAR to compute standard bases for global as well as for local orderings. One of his examples is the following:

Example: The irreducible curve C with affine equation $f(x,y)=0$,

$$\begin{aligned}f(x,y) = y^2 - 2y(x^{10} &+ \frac{1}{2}x^9y^2 - \frac{1}{8}x^8y^4 + \frac{1}{16}x^7y^6 - \frac{5}{128}x^6y^8 + \frac{7}{256}x^5y^{10} \\ &- \frac{21}{1024}x^4y^{12} + \frac{33}{2048}x^3y^{14} - \frac{429}{32768}x^2y^{16} + \frac{715}{65536}xy^{18} \\ &- \frac{2431}{262144}y^{20}) + x^{20} + x^{19}y^2\end{aligned}$$

has degree 21 and an A_{228}–singularity $(x^2 - y^{229} = 0)$ as its only singularity.

In order to verify this, one may proceed, using SINGULAR, as follows:

```
>ring s = 0,(x,y),ds;
>poly f = y2-2x10y-x9y3+1/4x8y5-1/8x7y7+5/64x6y9-7/128x5y11+21/512x4y13
        -33/1024x3y15+429/16384x2y17+x20-715/32768xy19+x19y2+2431/131072y21;
>matrix Hess = jacob(jacob(f));         //the Hessian matrix of f
>print(subst(subst(Hess,x,0),y,0));     //the Hessian matrix for x=y=0
0,0,
0,2
>vdim(std(jacob(f)));                  //the Milnor number of f
228
```

Since the rank of the Hessian at 0 is 1, f has an A_k singularity at 0; it is an A_{228} singularity since the Milnor number is 228. To show that the projective curve C defined by f has no other singularities, we have to show that **C has no further singularities in the affine part and no singularity at infinity.** The second assertion is easy.

The first follows from

$$\dim_{\mathbb{C}}(K[x,y]_{\langle x,y\rangle}/\langle \text{jacob}(f), f\rangle = \dim_{\mathbb{C}}(K[x,y]/\langle \text{jacob}(f), f\rangle,$$

confirmed by Singular:

```
>vdim(std(jacob(f)+f));
228                        //multiplicity of Sing(C) at 0
>ring r = 0,(x,y),dp;
>poly f = fetch(s,f);
>vdim(std(jacob(f)+f));
228                        //total multiplicity of Sing(C)
```

The **existence problem (II) for hypersurfaces in $\mathbb{P}^3$ of low degree** with specific singularities (such as nodes) has attracted attention of many researchers.

Let $m(d) :=$ maximum number of nodes on a surface X of degree d in $\mathbb{P}^3_{\mathbb{C}}$.

It is known: $m(d) = 1, 4, 16, 31, 65$ for $d = 2, 3, 4, 5, 6$.

For $d \geq 7$ we only know $\frac{5}{12}d^3 \leq m(d) \leq \frac{4}{9}d^3$ up to $O(d^2)$, but the exact value of $m(d)$ is unknown.

Note that the lower bounds are obtained in each case by a specific construction, due to Schläfli, Kummer, Togliatti, Chmutov and Barth.

In 2004 **O. Labs constructed a surface of degree 7 with 99 nodes** which is the current world record for surfaces of degree 7 (but which is still smaller than the known upper bound 104).

The construction of Labs is a very instructive example on how geometric reasoning with computer experiments over finite fields of small characteristic can be used to support research in algebraic geometry.

The arguments of Labs can be roughly summarized as follows. Inspired by previous work of Barth and Endraß, Labs considers a 6–parameter family $S_{a_1,\ldots,a_6} \in \mathbb{Z}[x, y, z]$ of homogeneous polynomials of degree 7, and the aim was to construct explicit algebraic numbers $a_1, \ldots, a_6$ such that $S_{a_1,\ldots,a_6}$ defines a nodal surface having more than the previously known 93 nodes. Computer experiments with SINGULAR over small prime fields suggested that the maximum number of nodes on $S_{a_1,\ldots,a_6}$ is 99 and that such examples should exist for $a_6 = 1$.

Using the symmetry of the family $S = S_{a_1,\ldots,a_5,1}$, it is sufficient to consider the plane curve defined by $S_y := S|_{y=0}$ and find parameters $\alpha_1, \ldots, \alpha_5$ such that S_y has many nodes (from which the number of nodes on S can be computed).

Of course, to work in the plane $y = 0$ allows much faster computations. By running SINGULAR computations over all possible parameter combinations for small prime fields $\mathbb{F}_p (11 \leq p \leq 53)$ he finds some 99–nodal surfaces over these fields. To find conditions for the parameters, Labs used geometric properties of the plane curve S_y together with extensive SINGULAR computations such as elimination and factorization.

He ended up with $a_1, \ldots, a_5$ being polynomial expression in $\alpha \in \mathbb{C}$, $7\alpha^3 + 7\alpha + 1 = 0$, such that the resulting polynomial S_α defines a surface with exactly 99 nodes over several prime fields.

It turns out that the same conditions give a 99–nodal septic surface in characteristic 0 which can be proved by a straightforward computations with SINGULAR.

The following surface in $\mathbb{P}^3(\mathbb{C})$ of degree 7 with equation $S_\alpha = P - U_\alpha$ has exactly 99 nodes and no other singularities, where

$$\begin{aligned}
P :\ &= x\cdot\left[x^6 - 3\cdot 7\cdot x^4y^2 + 5\cdot 7\cdot x^2y^4 - 7\cdot y^6\right] \\
&\quad +7\cdot z\cdot\Big[\big(x^2+y^2\big)^3 - 2^3\cdot z^2\cdot\big(x^2+y^2\big)^2 + 2^4\cdot z^4\cdot\big(x^2+y^2\big)\Big] - 2^6\cdot z^7, \\
U_\alpha :\ &= (z + a_5 w)\left((z+w)(x^2+y^2) + a_1 z^3 + a_2 z^2 w + a_3 z w^2 + a_4 w^3\right)^2,
\end{aligned}$$

$$\begin{aligned}
a_1 &= -\tfrac{12}{7}\alpha^2 - \tfrac{384}{49}\alpha - \tfrac{8}{7}, & a_4 &= -\tfrac{8}{7}\alpha^2 + \tfrac{8}{49}\alpha - \tfrac{8}{7}, \\
a_2 &= -\tfrac{32}{7}\alpha^2 + \tfrac{24}{49}\alpha - 4, & a_5 &= 49\alpha^2 - 7\alpha + 50. \\
a_3 &= -4\alpha^2 + \tfrac{24}{49}\alpha - 4, & &
\end{aligned}$$

Note that $7\alpha^3 + 7\alpha + 1 = 0$ has one real solution $\approx -0,14010685$ and for this value all 99 nodes of S_α are real, which allows to draw a nice picture of S_α.



Lab's 99–nodal septic

The following SINGULAR code verifies that Lab's septic has indeed 99 nodes and no other singularities.

```
>LIB "all.lib";

>ring r = (0,alpha), (x,y,w,z), dp;
>minpoly = 7*alpha^3 + 7*alpha + 1;
```

```
>poly a(1) = -12/7*alpha^2 - 384/49*alpha - 8/7;
>poly a(2) = -32/7*alpha^2  + 24/49*alpha - 4;
>poly a(3) =    -4*alpha^2  + 24/49*alpha - 4;
>poly a(4) =  -8/7*alpha^2   + 8/49*alpha - 8/7;
>poly a(5) =    49*alpha^2      - 7*alpha + 50;

>poly P = x*(x^6-3*7*x^4*y^2+5*7*x^2*y^4-7*y^6)
 +7*z*((x^2+y^2)^3-2^3*z^2*(x^2+y^2)^2+2^4*z^4*(x^2+y^2))-2^6*z^7;
>poly C = a(1)*z^3+a(2)*z^2*w+a(3)*z*w^2+a(4)*w^3+(z+w)*(x^2+y^2);
>poly U = (z+a(5)*w)*C^2;
>poly S = P-U;
```

The following computation verifies that the total Tjurina number of S_α is 99 and that all singularities are ordinary double points, using the Hessian criterion. We check the total Tjurina number of the projective surface:

```
>ideal sl = jacob(S);          //the singular locus of S
>ideal newsl = groebner(sl);  //a groebner basis
>dim(newsl)-1;          //dimension of the projective variety.
0
>mult(newsl);           //total tjurina number
99
```

Check now that all singularities are ordinary double points:

```
>matrix mHS = jacob(jacob(S));
>ideal nonnodes = minor(mHS,2), sl;  //the ideal of non-nodes
>nonnodes = groebner(nonnodes);
>dim(nonnodes);
0
```

Since the dimension is zero, the projective dimension of nonnodes is -1, that is, there are no non-nodes.

References

[1] T. Bandman, G.-M. Greuel, F. Grunewald, B. Kunyavskii, G. Pfister, E. Plotkin: Engel-Like Identities Characterizing Finite Solvable Groups. Preprint, http://arXiv.org/abs/math.GR/0303165, (2003). To be published in Compos. Math. (2006).

[2] Bandman, T.; Greuel, G.-M.; Grunewald, F.; Kunyavskii, B.; Pfister, G.; Plotkin, E.: Two-variable identities for finite solvable groups. *Comptes Rendus de l'academie des sciences, Ser. I*, **337**, 581–586 (2003).

[3] Decker, W.; Lossen, C.: *Computing in Algebraic Geometry. A Quick Start using* SINGULAR. Springer-Verlag, 2006.

[4] G.-M. Greuel, Ch. Lossen, E. Shustin: Singular Algebraic Curves. Monograph to appear. Springer–Verlag.

[5] Greuel, G.-M.; Pfister, G.: *A* SINGULAR *introduction to commutative algebra.* Springer-Verlag, 2002.

[6] Greuel, G.-M.; Pfister, G.; Schönemann, H.: SINGULAR 3.0. A Computer Algebra System for Polynomial Computations. ZCA, TU Kaiserslautern (2005). `http://www.singular.uni-kl.de`.

[7] Labs, O.: A Septic with 99 real Nodes, arXiv:math.A6/0409348

[8] Pfister, G.; Schönemann, H.: Singularities with exact Poincaré complex but not quasihomogeneous. *Revista Mathematica Madrid*, **2**, 161–171 (1989).